

Connecting Your Output: Using Drill-down Techniques to Link Reports, Graphics, and Tables

Richard O. Smith, Data Explorations, Carlsbad, California
Arthur L. Carpenter, Data Explorations, Carlsbad, California

ABSTRACT

When displaying the output from your SAS® programs, it is no longer necessary to force your readers to go through a stack of output linearly from front to back. Using the Output Delivery System, ODS, and drill-down techniques allow you to produce reports, graphics, tables and other forms of output that automatically link to other related tables and graphics. If your reader wants to see the supporting statistics, or even the raw data, for a subgroup that looks a bit 'odd', there is no longer a need to have them flip pages to some appendix of your report. The link to the appropriate place in the appropriate supporting table can automatically be built into each of your tables and graphs.

Learn how you can easily connect these tables, reports, and graphics so that they can be used as links in the chain of data display items.

INTRODUCTION

Reports and output from computer programs have historically been printed in hard-copy form. This was also true for SAS® programs until recently with the inception of the Output Delivery System (ODS) and html technology. It is now no longer necessary to force your readers to go through stacks of output linearly from front to back. The use of ODS and drill-down techniques allow you to produce reports, graphics, tables and other forms of output that automatically link to other related tables and graphics.

If your reader wants to see the supporting statistics, or even the raw data, for a subgroup that looks a bit 'odd', there is no longer a need to have them flip pages to some appendix of your report. The link to the appropriate place in the appropriate supporting table can automatically be built into each of your tables and graphs.

We will examine how to create links from print, report, statistical and graphics procedures. We will also examine how to create a linkable index to separate output into groups.

Our children are growing up in the computer age and many are very literate in the use of computers and software. They will have a great advantage in the understanding and the comfort in using computers and software in the future. Not only will future users of software easily understand links and drill-down techniques, they will expect it.

KEYWORDS

Output Delivery System, ODS, HTML, WEBFRAME, Macro, Drill-down

OUTPUT DELIVERY SYSTEM – THE UNDERLYING TOOL

The Output Delivery System is the underlying tool that allows us to easily create output in a wide variety of formats from virtually all SAS procedures. Many fine publications and user-group presentations are available on the use of ODS.


The ODS HTML output format allows tables and graphs with drill-down capability, interactive manipulation, and creation of linked Pages or Tables of Content.

When a series of graphs or tables are to be added to an HTML file, additional options can be used to build a linked Table of Contents for the various components allowing one to jump quickly to the selection. In the following example the FRAME=, CONTENTS=, and PAGE= options are added to the ODS HTML statement.

```
* ODS HTML coordination using FRAME;  
ods listing close;  
ods html path = "&drive\&project\tables\  
          body = 'multiple.html'  
          contents= 'multcontents.html'  
          page = 'multpage.html'  
          frame = 'multframe.html'  
          (title='Magdata Analysis');  
  
ods proclabel 'Data Summary';  
proc univariate data=sasclass.magdata;  
  var ampida ampidb;  
  title1 'Magdata Summary';  
run;
```

```
ods proclabel 'Distribution';
proc gchart data=sasclass.magdata(where=(ampida>100));
  hbar Ampida;
  title2 'Ampida Distribution';
run;
quit;
ods html close;
```

When the file named by the FRAME= option is browsed, the user sees a display such as the left one shown below. This allows the individual objects created by the UNIVARIATE and GCHART procedures to be selected directly.



A number of ODS statements and options are available to enhance this type of output file. Of these, the ODS PROCLABEL statement and the TITLE= option are shown.

One can also create thumbnails using the WEBFRAME goptions device that will provide thumbnail images that are linked to the full-size image. The following code using WEBFRAME generates the graphics output seen above right:

```
filename regplt "&drive\&project\results\regsctr";

goptions device=webframe gsfname=regplt;
symbol1 c=blue v=dot r=45;
axis1 label=(a=90 f=simplex c=blue 'Adjusted Rate');
title2 'County Data Plots';
proc gplot data=sasclass.cnty;
.... more code ....
```

BUILDING AN INDEX WITH PROC PRINT

In the previous examples, a series of graphs were built using the WEBFRAME device. Each execution of the GREPLAY wrote another series of graphs to another directory. The user can now display all of the graphs in a given subdirectory simply by going to that directory, finding the index.html file and browsing it. This is still quite a challenge if there are numerous directories and/or if there are many graphs within any given directory. What we next need is a single file that, when browsed, will direct the user to the correct index file / directory combination by a simple click of the mouse. We will create this file with a simple PROC PRINT.

The data set to be printed must contain a variable whose value is a valid HTML anchor tag that references the name of another file. The form of this link (GRPREF in the following DATA step) includes the value to be displayed (the report group in this case which is stored in CRPTGRP), and the name/location of the file to branch to (the link is stored in the variable MYLINK).

```
data prep2;
  set prep1;
  length grpref mylink $150 crptgrp $3;

  *rptgrp+1;
  crptgrp= trim(left(put(rptgrp,3)));
  myLink = "\\&project\hazardratios\hr" || trim(crptgrp) || '\index.html';
  grpref = "<a HREF=" || trim(myLink) || '>' || trim(crptgrp) || '</a>';
  label grpref = 'Report Group';
run;

* Define location and index name for new index;
ods html path="&drive\&project\&outloc" (url=none)
  body="&indexname..html";
```

```

%put going to "&drive\&project\&outloc";
ods listing close;

proc print data=prep2 label noobs;
  var grpref groupdesc;
  title1 "&title Control File: &indsn";
  run;
ods html close;

```

When the file generated by the PROC PRINT is displayed, only the portion of the variable GRPREF that came from the variable CRPTGRP will be displayed. When the user selects one of these values, the corresponding INDEX.HTML file, which was built by WEBFRAME, will be displayed.

In a more complex study or analysis presentation more than one level of PROC PRINT indexes may be needed. Perhaps the first level selects general types of reports or general areas of interest. A selection here then points to a secondary index that allows the reader to further select the set of graphs of interest. Effectively this process creates a Table of Contents that passes the reader directly to the graphs of interest.

CREATING DRILL-DOWN GRAPHS AND CHARTS

Not only do we want to be able to use the tools discussed above to locate and display a specific graph, we may also want to use the graph itself to point to more information. As we browse the graph created using the techniques shown above, we want to be able to create 'hot zones' that allow us to drill down through a portion of a graph by clicking on a bar, line, or symbol, so that we can then display and browse another related graph or table.

In the example below both a histogram and a scatter plot are generated as GIF files. Each of these graphs contains information that is STATION specific. When browsing, if the cursor is moved over a station specific portion of the graph, a 'hot' zone, it changes to a pointer. The pointer indicates that clicking on that hot zone will cause a branch to another file or graph. In this case the branch will be to a table created by PROC PRINT of that station's data.

First we create a PROC PRINT for each value of the variable STATION. Notice that the name of the HTML file, the TITLE, and the value of the variable STATION in the WHERE clause all contain the name of the station (AZU in this example).

```

* List the AZUSA data;
ods html path="&drive\&project\figures"(url=none)
  body='azu.html';
proc print data=sasclass.ca88air(where=(station='AZU')) noobs;
  var month co o3 no3 tem;
  title1 'AZU Pollution';
  footnote;
  run;
ods html close;

```

In the production environment it is likely that the list of stations will be stored in a macro array. The PROC PRINT can then be placed in a %DO loop. This allows the automated generation of the secondary files as shown below.

```

%do i = 1 %to &stacnt;
  * List the &&sta&i data;
  ods html path="&drive\&project\figures"(url=none)
    body='&&sta&i.html';
  proc print data=sasclass.ca88air(where=(station="&&sta&i")) noobs;
    var month co o3 no3 tem;
    title1 "&&sta&i Pollution";
    footnote;
    run;
  ods html close;
%end;

```

The key to pointing to the secondary files (created by PROC PRINT above) is a character variable that contains a reference file pointer. The form of the value of this variable will be 'href=xxxxxx.html', where xxxxxx is data dependent (in this case the station name). In the data set to be plotted we create a variable, DRILLSTA, which contains the name of the station as part of the file pointer.

```

data ca88air;
  set sasclass.ca88air;
  length drillsta $15;
  drillsta = 'href='||trim(left(station))||'.html';
  run;

```

This variable is utilized by the PROC GCHART in the VBAR statement through the HTML= option. Notice also that we are using the GIF device, although other devices, such as, WEBFRAME which was shown above, could also be used.

```
goptions device=gif;
ods html path="&drive\&project\figures"(url=none)
         body='chart.html';

PROC GCHART DATA=ca88air;
  VBAR station / type=mean sumvar=o3
  patternid=midpoint
  subgroup=station
  html=drillsta
  raxis=axis1 maxis=axis2;
run;
quit;
ods html close;
```

The SUBGROUP= option is also sometimes required with VBAR and HBAR chart types in order to make use of the HTML= option. This is true for this chart even though the SUBGROUP= option is not otherwise needed. It is also possible to create hot zones in a scatter plot. In the following example the same data that was used above is also plotted. In this case there is one line per station. Because the DRILLSTA variable is a constant for each station, each line becomes a separate hot zone.

```
goptions device=gif;
ods html path="&drive\&project\figures"(url=none)
         body='gplot.html';

proc gplot DATA=ca88air;
  plot o3*month=station /
  html=drillsta
  htmllegend=drillsta
  vaxis=axis1;
run;
quit;
ods html close;
```

The HTMLLEGEND= option creates hot zones in the legend as well. In this example the variable DRILLSTA is constant for each station. If instead it had been unique for each plotted point, then each point could have been a separate hot zone.

USING NON-STATIC DRIVERS

Device drivers such as ACTIVEX and JAVA allow you to create graphs that are not static. The user of the graph can manipulate these graphs as it is browsed.

ACTIVEX

The ACTIVEX driver allows the developer to build graphs that can be changed and adapted by the user outside of SAS/GRAPH. Unlike the GIF, WEBFRAME, and HTML devices this device creates an HTML file, not a GIF file.

In order to control the size of the graph and to prevent it from not fitting on the browser screen, the XPIXELS= and YPIXELS= options are often needed. The values taken on by these two options will depend on the display resolution used by the person browsing the resulting HTML file.

```
goptions reset=all border ftext=simplex;
goptions device=activex
         xpixels = 640
         ypixels = 480;
pattern1 c=cyan;
pattern2 c=yellow;
pattern3 c=red;

ods html body="&drive\&project\figures\SummerO3.html";



PROC GCHART DATA=sasclass.ca88air;
  where month in(6 7 8);
  VBAR3D station / type=mean sumvar=o3
  shape=cylinder
  cframe=pink
  group=month
```

```

patternid=midpoint;
title1 'Average Summer Ozone Concentrations';
title2 'Three Dimensional Chart Elements';
title3 a=90 f=simplex h=1 'Parts per Million';
footnotel h=1 f=simplex 'Summer Months';
run;
quit;
ods html close;

```

This code generates the graph below left.



JAVA

The JAVA driver builds the JAVA components so that when the HTML file is browsed the appropriate JAVA Applets construct the graph. Like the ACTIVEX graphs these graphs are not static to the user and can be reshaped, resized, and otherwise greatly modified. As in the example for the ACTIVEX device, notice the use of the XPIXELS= and YPIXELS= options to control the initial size of the graph.

The PARAMETERS= option in the ODS HTML statement is used to place parameters and associated values into the JAVA code. In this case the DRILLDOWNMODE parameter receives a value of LOCAL. The following code generates the graph above right.

```

goptions reset=all border ftext=simplex;
goptions device=java
          xpixels = 600
          ypixels = 600;
pattern1 c=cyan;
pattern2 c=yellow;
pattern3 c=red;

ods html file="&drive\&project\figures\SummerMonth.html"
          parameters=( "DRILLDOWNMODE"="LOCAL" );

PROC GCHART DATA=sasclass.ca88air;
  where month in(6 7 8);
  VBAR3D station / type=mean sumvar=o3
          shape=cylinder
          cframe=pink
          group=month
          patternid=midpoint;
  title1 'Average Summer Ozone Concentrations';
  title2 'Three Dimensional Chart Elements';
  title3 a=90 f=simplex h=1 'Parts per Million';
  footnotel h=1 f=simplex 'Summer Months';
run;
quit;
ods html close;

```

SUMMARY

The development of ODS has provided the capability to create links, drill-down, and interactive graphics. Large numbers of graphs that are interrelated with tables and other data displays can be difficult in the production environment. Not only is coordination required between linked graphs and tables, but just locating and pointing to the myriad of combinations can be problematic. Through the use of index tables and graphs with drill-down hot zones it is possible to set up an environment that is much more easily navigated by the user.

A number of graphics device drivers are available which enable the programmer to create a variety of types of graphs that can be browsed by the user. When these are coupled with ODS and controlled through the use of macros a very robust and expandable system can be created.

Tie these techniques together and you can generate a series of interconnected tables and graphs in an automated user-friendly environment. Reports can now be browsed interactively instead of linearly through large stacks of output.

The poster presentation for this paper will have many examples of links, drill-down graphs and tables. Text in the poster will be limited descriptive information about each example.

ABOUT THE AUTHORS

Richard Smith and Art Carpenter are senior partners at Data Explorations. Data Explorations, a SAS Alliance Affiliate Member™, provides data management, analyses, and SAS programming services nationwide.

Richard O. Smith

Richard Smith has a Masters in Biology/Ecology and has provided complete data management and analysis services for numerous environmental research projects as a senior biologist, SAS programmer, and project manager. He has also provided programming and management services for the health related industries. He has been using SAS extensively since 1981 and is a SAS Certified Professional™.

Arthur L. Carpenter

Art Carpenter is a SAS Certified Professional™. His publications list includes three books on SAS topics (*Annotate: Simply the Basics*, *Quick Results with SAS/GRAPH® Software*, and *Carpenter's Complete Guide to the SAS® Macro Language*), two chapters in *Reporting from the Field*, and numerous papers and posters presented at various user group conferences. Art has been using SAS since 1976 and has served in a variety of positions in user groups at the local, regional, and national level.



AUTHOR CONTACTS

Data Explorations

2270 Camino Vida Roble, Suite L
Carlsbad, CA 92009

Richard O. Smith

(760) 438-1336
RSmith@SciX.com

Arthur L. Carpenter

(760) 945-0613
Art@Caloxy.com

TRADEMARK INFORMATION

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.